MS 101: Makerspace
EE Lecture: 03
Electronic Devices and Circuits II

Joseph John, Dinesh Sharma,
B.G. Fernandes, Debraj Chakraborty,
Kishore Chatterjee, P.C. Pandey,
Narendra S. Shiradkar and Kushal Tuckley

EE Department
IIT Bombay, Mumbai

Academic Year: 2023-2024, Semester: II (Spring)

DC Power Supplies

- Unregulated Power Supplies:
  - Half-Wave Rectifier
  - Full-wave Rectifier: Bridge rectifier circuit
- Regulated Power Supplies:
  - Zener regulation
  - 3 Terminal power supply regulators

Half-wave Rectifier





- Diode conducts when the top of the secondary winding is positive.
- The diode is blocked when the top of secondary winding is negative.
- The output has only the positive peak with voltage reduced by a diode drop.
- The output has a non-zero DC component, but it is not usable directly as a DC supply for electronic circuits.

Simulating the Half-wave Rectifier






```
*Diode Rectifier
*For 12V RMS, peak voltage is
*12*1.414 = 16.97V
Vin TxTop 0 sin(0 16.97 50 0 0)

D1 TxTop Out DF
.MODEL DF D (IS=6.22n N=1.9224
+ RS=0.336 CJ0=764f VJ=0.75
+BV=200 TT=2.88n)
RL Out 0 1k
*Transient Analysis
.tran 100u 80m 20m
.control
run
plot V(TxTop) V(Out)
plot -I(Vin)
.endc
.end
```

Results of Simulation





Full-wave Bridge Rectifier



- A full-wave bridge rectifier uses 4 diodes.
- The bottom terminal of transformer secondary is **not grounded**.
- Two diodes conduct and two are blocked in each half cycle of AC input.
- Both half cycles produce a positive output. Therefore this is called a full wave rectifier.
- The peak output voltage is less than the the peak input voltage by two diode drops.

Full-wave Bridge Rectifier: +ve Half Cycle



- Diodes D1 and D4 conduct, while D2 and D3 are blocked.
- The bottom terminal of transformer secondary (driven -ve by transformer) goes to $-V_D$.
- The top terminal of transformer goes to $V_{ac} - V_D$.
- Therefore the output voltage goes to $V_{ac} - 2V_D$.
- The peak output voltage is less than the the peak input voltage by two diode drops.

Full-wave Bridge Rectifier: -ve Half Cycle



- Diodes D2 and D3 conduct, while D1 and D4 are blocked.
- The top terminal of transformer secondary (driven -ve by transformer) goes to $-V_D$.
- The bottom terminal of transformer goes to $V_{ac} - V_D$.
- Therefore the output voltage goes to $V_{ac} - 2V_D$.
- The peak output voltage is less than the the peak input voltage by two diode drops.

Full-wave Bridge Rectifier Output



- Output is $+$ve during positive as well as negative cycles.
- The output voltage magnitude is two diode drops less than the input.
- The output is driven during the whole AC cycle. The repetition frequency at the output is double the frequency of the sinusoidal input.
- Full wave rectification produces smoother DC output compared to Half wave rectifier.

Simulating the Full-wave Bridge Rectifier




```
*Bridge Rectifier
D1 TxTop Out DF
D2 TxBottom Out DF
D3 Gnd TxTop DF
D4 Gnd TxBottom DF
RL Out TestPt 1k
Vtest TestPt Gnd 0V
```

```
*Full wave Bridge Rectifier
*For 12V RMS, the peak value is 12*1.414 = 16.97V
Vin TxTop TxBottom DC 0
+ Sin(0 16.97 50 0 0)
.MODEL DF D (IS=6.22n N=1.9224
+ RS=0.336 CJ0=764f VJ=0.75
+ BV=200 TT=2.88n)

.tran 100u 80m 20m .control
run
plot V(TxTop) V(TxBottom)
plot V(TxTop)-V(TxBottom) V(Out)
plot -I(Vin) I(Vtest)
.endc
.end
```

Results of Simulating Full Wave Bridge Rectifier





Producing DC voltage from Rectifiers

- As we have seen, half and full wave rectification produce outputs with non-zero DC component. However, the output voltage varies substantially with time.
- We know that a capacitor does not permit fast changes of voltage across its terminals.
- We can convert rectifier outputs to reasonably constant DC values by putting large capacitors across their output.
- Large value capacitors (with values of ten to several thousand μF) are made as electrolytic capacitors, which are polar.
- Great care should be taken when using polar capacitors such that the correct polarity voltage is applied to its terminals.

Half Wave Rectifier with Capacitor

The circuit on the right shows a half-wave rectifier with a large capacitor at its output.



- During the positive cycle, the diode is forward biased and charges the capacitor to the peak voltage minus a diode drop.
- As the input voltage passes its peak value, it begins to reduce. Once it drops below the voltage across the charged capacitor by a diode drop, the diode is blocked and the capacitor discharges through the load resistor.
- If the capacitor value is large, the output voltage will not drop substantially during this cycle when the diode is blocked.
- The capacitor will be charged again when the input voltage rises above the reduced voltage across the capacitor (plus a diode drop), during the positive cycle.

Effect of $R_L C$ Time constant

When $R_L \to \infty$, the time constant $R_L C$ also $\to \infty$.
Once the capacitor charges up, there is no discharge and the output voltage remains constant.



- For finite $R_L C$, the capacitor discharges during the time that the input is less than the output voltage by a diode drop $(T - \delta T)$.
- Once the input voltage exceeds the reduced output voltage by more than a diode drop, it charges up (for time δT).



Periodic charge and discharge of the capacitor causes the output voltage to fluctuate. This is called the ripple voltage.

Ripple dependence on C

- C charges during δT and discharges during $T - \delta T$.
- Ripple voltage V_r depends on the rate of capacitor discharge. It increases with i_L (load current).



Ripple voltage could be decreased by increasing C. However it is not a good solution because the current during initial charging of C may be too large.



The half-wave rectifier with C is very seldom used due to its higher ripple voltage

Full Wave Rectifier with Capacitor

The circuit on the right shows a full-wave rectifier with a large capacitor at its output.



- The bridge rectifier output is positive in both cycles of input AC. The output can charge the capacitor to the AC peak value $-2V_D$ in either AC cycle.
- As the rectified voltage passes its peak value, it begins to reduce. Once it drops below the voltage across the charged capacitor by two diode drops, all diodes are blocked and the capacitor discharges through the load resistor.
- If the capacitor value is large, the output voltage will not drop substantially till the next charge up interval, which now occurs in positive as well as negative cycles of input AC.

Ripple Voltage for a Full Wave Rectifier





- The bridge rectifier output is positive in both cycles of input AC. The output can charge the capacitor to the (AC peak value $-2V_D$) in either AC cycle.
- Compare to half wave rectifier, the discharge interval is about half. Therefore for the same value of smoothing capacitor and load current, the ripple voltage is half as much.
- A larger capacitor will give lower ripple. However, the maximum value of C depends on the maximum inrush current which can be supported by the diode. (Inrush current is the initial current required to charge the capacitor from a totally discharged state).

Simulating a Full Wave Rectifier with C

We use the same input file as was used for the bridge rectifier, but with the addition of a capacitor between output and ground.

Figure on the right shows the input voltage to the bridge rectifier (in red) and the output voltage (in blue).

The load resistor used for simulation was 1 kΩ and the smoothing capacitor was 500μF



Ripple and Inrush current

Simulation results shown here are for a full wave bridge rectifier with a load resistor of 1 kΩ and a smoothing capacitor of 500μF.



The plot on the left uses a magnified scale to show ripple. There is about 250 mV of ripple on the output.

Plot on the right shows the current drawn from transformer secondary.



Notice the high value of inrush current(about 2.3 A!) compared to repetitive charging current pulses of about 150 mA per cycle.

Ripple and Inrush current: $R_L = 500\Omega$, $C = 10\mu F$

To show the effect of the smoothing capacitor on ripple and inrush current, we show results for 3 values of smoothing capacitors with $R_L = 500\Omega$. This set is for C= 10μF.



Transformer secondary voltage (red) and output voltage (blue).



The output shows a ripple of about 10V! This is clearly unacceptable.



Inrush current and repetitive charging current have the same value of about 60 mA.

Ripple and Inrush current: $R_L = 500\Omega$, $C = 100\mu$F



Transformer secondary voltage (red) and output voltage (blue).



The output shows a ripple of about 2.1V. This is quite high.



Inrush current is about 500 mA while repetitive charging current is of about 220 mA.

Ripple and Inrush current: $R_L = 500\Omega$, $C = 1000\mu F$



Transformer secondary voltage (red) and output voltage (blue).



The output shows a ripple of about 250 mV. This is appreciable, still.



Inrush current is $\approx$ 4.2A while repetitive charging current is $\approx$ 300 mA.

Problems with Unregulated Power Supplies

- Simulation data shows the trade off one must make between acceptable ripple voltage and Inrush current.
- The output voltage varies with changes in input voltage. We would ideally like the output voltage to stay constant even when the input voltages changes. (Variation of output voltage with input voltage is known as “Line Regulation”).
- The output voltage and ripple also depend on the load current. Again, we would like the output voltage to remain unchanged with variations in load current. (Variation of output voltage with load current is characterised as “Load Regulation”).
- To get around these problems, we use specific circuits to get outputs with low ripple and good Line and Load regulation. Supplies using these circuits are known as regulated power supplies.

Regulated Power Supplies

Two main approaches for making electronically regulated power supplies are:

1. Zener diode based regulated supply
2. Voltage regulator based supplies using negative feedback

Both these circuits use the output of an unregulated supply as their input.

Zener based circuits were discussed in the lecture on diodes. We shall describe and simulate this circuit to look at its line and load regulation properties.

Zener Diodes for Regulated Supplies

- Zener breakdown occurs due to quantum mechanical tunnelling of carriers between heavily doped p and n regions of a diode. Most diodes termed as “Zener diodes” actually do not use the Zener breakdown phenomenon, and the term is, in fact, misleading. However, it is widely used and we shall also apply the term according to its popular usage.
- “Zener” diodes actually use avalanche break down, which occurs when a diode is subjected to high reverse bias.
- When we exceed the breakdown field of a diode, carriers are accelerated by the high field and acquire high kinetic energy. When these collide with an atom, they dislodge bound electrons and thus create new electron-hole pairs.
  These newly created electron-hole pairs are also accelerated by the field and knock out even more electrons. This process generates an avalanche of carriers and results in a sudden increase in the leakage current.

Zener Diodes for Regulated Supplies

The figure on the right shows the I-V characteristics of a Zener diode.

- In the avalanche breakdown region, the voltage across the diode shows very little change even as the current spans several decades of change.
- This can be used in a circuit to generate a stable output voltage.



Zener Regulated Supply

The figure on the right shows a simple circuit for using a Zener diode to regulate the voltage. R_S limits the current through the Zener diode when it breaks down.



Let the break down voltage of the Zener diode be V_Z. For $V_{out} < V_Z$, the Zener diode draws negligible current. R_s and R_L form a potential divider, such that $V_{out} = V_{in} R_L/(R_S + R_L)$.
Correspondingly, $V_{in} = V_{out}(R_S + R_L)/R_L$ for $V_{out} \leq V_Z$.

When the input voltage exceeds $V_Z(R_S + R_L)/R_L$, the Zener diode breaks down and draws heavy current. This additional current causes additional drop across R_S and keeps the output voltage at $\approx V_Z$.

Thus the output voltage remains reasonably constant while the input voltage can vary over a large range with $V_{in} > V_Z(R_S + R_L)/R_L$.

Simulating the Zener Regulated Supply

```
*Zener Regulator
Vin Supply Gnd 0
*Series limiter
RS Supply Vout 300
*Load resistor
RL Vout Gnd 1000
```



Output Voltage



Zener current

```
*Zener modeled as a diode with
*low breakdown voltage of 4.7V
D1 Ztest Vout DF
.MODEL DF D (IS=6.22n N=1.9224
+RS=0.336 CJ0=764f VJ=0.75
+BV=4.7 TT=2.88n)
*0V source inserted to measure
* Zener current
VZtest Ztest Gnd 0
```

```
*DC analysis: Sweep input voltage
.DC Vin 0.0 15.0 0.05
.control
run
plot V(Vout)
plot I(VZtest) ylimit 0 5.5m
.endc
.end
```

Voltage Regulation with Feedback

- It is possible to use electronic circuits which monitor the output voltage and counteract any deviation from the desired voltage by adjusting the conduction through internal devices.
- These regulators can be made using a combination of discrete devices and integrated circuits. However, these days special purpose integrated circuits have become available which makes the job of making electronically regulated power supplies quite easy.
- We shall not describe or analyse these circuits. However, we'll describe how to use such ready-made circuits.
- In particular, integrated circuits known as 3 terminal regulators are very convenient to use. These are available with type numbers 78xx for commonly used positive output voltages and 79xx for commonly used negative output voltages. Here xx is the desired output voltage.

Regulated Power Supply using a 3 Terminal Regulator

Specifications:

$14.5\text{ V} \leq V_{in} \leq 30\text{ V}$

$11.5\text{ V} \leq V_{out} \leq 12.5\text{ V}$

$I_L \leq 1\text{ A}$





- The input voltage is generated by an unregulated power supply – typically a step down transformer, bridge rectifier and smoothing capacitor.
- Power dissipated by the IC is: $(V_{in} - V_{out}) \times I_L$.
  This can be substantial – for example, if $V_{in} = 20$ V, $I_L = 0.8$ A, the package needs to dissipate $(20 - 12) \times 0.8 = 6.4$ W!.
- The package for 7812 has a metallic tab internally connected to ground. This can be bolted to a heat sink if required.

Internal Circuit of 7812 Voltage Regulator



Major blocks of the 7812 IC:

- Zener reference for comparison
- Error Amplifier
- Series pass transistor (Q16) (This is the controlled conductive element)
- Short circuit protection.

Source: 7812 Data sheet, National Semiconductor Corp., 2000.

Features of an IC regulator

- Excellent line and load regulation:
  Output voltage shows very little variation while the input voltage and load current vary over a large range.
- Minimum input voltage should be typically about 2 to 3 V above the output voltage. This is knows as the “head room” voltage.
- A small capacitor – typically 0.1 μF to 1 μF is put at the output (close to the IC leads). This is to prevent oscillations.

The regulator IC uses an error amplifier with negative feedback. Because of the delay through this amplifier, the negative feedback may become positive feedback at some frequency. This may lead to oscillations.

Widely available 3 Terminal Voltage Regulator ICs

- For positive output voltages:
  1. 7805 for $V_{out} = 5$ V.
  2. 7806 for $V_{out} = 6$ V.
  3. 7809 for $V_{out} = 9$ V.
  4. 7812 for $V_{out} = 12$ V.
- For negative output voltages:
  1. 7905 for $V_{out} = -5$ V.
  2. 7906 for $V_{out} = -6$ V.
  3. 7909 for $V_{out} = -9$ V.
  4. 7912 for $V_{out} = -12$ V.

  Input voltage needs to be negative for negative output voltages.

There are many other regulator ICs available.

For example, it is possible to generate a negative output voltage from a positive unregulated voltage. ICs which provide this capability internally charge a capacitor and switch its leads to generate the negative input voltage. This is followed by a regular negative voltage regulator.